SOFTWARE CONTROL OF THE DISK II OR IWM CONTROLLER

PREPARED BY

N L

APRIL 26, 1984 ·

REVISION 1, MAY 10, 1984

SOFTWARE CONTROL OF THE DISK II OR IWM CONTROLLER

Each of the eight expansion slots of the Apple II computer has the exclusive use of sixteen memory locations for I/O control. These memory locations can be used as software switches. A software routine can instruct the interface card in a particular slot to perform a predefined hardware task by toggling a software switch. Whenever the Apple II addresses one of the sixteen I/O locations allocated to a particular slot, the signal on pin 41 of that slot, called DEVICE SELECT/, switches to the active low state. This signal, in conjunction with the four low-order address lines (AO - A3), can be used to enable logic in the peripheral card to perform a particular task. The following table illustrates the memory space for the sixteen I/O locations for each expansion slot.

TABLE 1: PERIPHERAL CARD I/O SPACE

SLOT	LOCATIONS
0	\$C080 - \$C08F
1	\$C090 - \$C09F
2	\$COAO - \$COAF
3	\$COBO - \$COBF
4	\$COCO - \$COCF
5	\$CODO - \$CODF
6	\$COEO - \$COEF
7	\$COFO - \$COFF

One common method of accessing peripheral card I/O control softswitches through software is to use the Apple's (6502) indexed addressing mode. For example, a LDA \$C080, X instruction can be used to access softswitch O in any slot, if the index register X is loaded with a value equal to the slot number times sixteen.

Sixteen peripheral I/O addresses are used to control the functions of the disk II controller. The following table illustrates the functions of the sixteen software switches.

TABLE 2: DISK II CONTROLLER SOFTSWITCHES

ADDRESS	FUNCTION
\$C080, X	PHASE O OFF
\$C081, X	PHASE O ON
\$C082, X	PHASE 1 OFF
\$C083, X	PHASE 1 ON
\$C084, X	PHASE 2 OFF
\$C085, X	PHASE 2 ON
\$C086, X	PHASE 3 OFF
\$C087, X	PHASE 3 ON
\$C088, X	TURN MOTOR OFF
\$C089, X	TURN MOTOR ON
\$C08A, X	SELECT DRIVE 1
\$C08B, X	SELECT DRIVE 2
\$C08C, X	Q6L
\$C08D, X	Q6H
\$COBE, X	Q7L
\$COBF, X	Q7H

The index register X has the value of slot number times 16. The last four addresses have the following functions.

<u>96</u>	97	FUNCTION
L	L	READ
H	L	SENSE WRITE PROTECT OR PREWRITE STATE
L	H	WRITE
H	H	WRITE LOAD

In general, any valid 6502 instruction can be used to access the above soft-switch address, except a load instruction is used to read a byte of encoded data from the controller and a store instruction is used to write a byte of encoded data to the controller. Below are typical examples demonstrating the use of the disk II controller softswitches. It is assumed that both Q6 and Q7 are low at the beginning of the read/write examples.

SELECT DRIVE

LDA	\$C08A,	X	SELECT	DRIVE	1
LDA	SCO8B.	X	SELECT	DRIVE	2

The hardware design of the controller allows only one drive to be selected at one time. A LDA \$CO8A, X instruction will select drive 1 and deselect drive 2. A LDA \$CO8B, X instruction will select drive 2 and deselect drive 1.

MOTOR ON

LDA	\$C089,	X	TURN	MOTOR	ON
LDA	\$C088,	X	TURN	MOTOR	OFF

It should be noted that there is only one interface signal (ENABLE/), going from the controller to each floppy disk drive, which is used to enable the drive's read/write function and to turn on the motor. Both the select drive and the motor on instructions must be executed in order to activate the ENABLE/ signal of a particular drive. A typical program will select the drive first and then turn on the motor at a later time. After the completion of the motor on instruction, the program should wait at least 1 second for the motor to come up to speed, before read/write functions can be performed reliably.

The disk II controller hardware will keep the ENABLE/ signal to its active low state for approximately one second after the execution of the motor off instruction, therefore read/write can be performed reliably within this period. To be on the safe side, the program should verify that the motor is spinning by monitoring the change in data pattern read from the drive. This delay in turning off the motor facilitate rapid and repeat access to the same drive.

SENSE WRITE PROTECT

LDA	Q6H, X	WRITE PROTECT SENSE MODE
LDA	Q7L, X	READ CONTROLLER STATUS REG.
BMI	WRPROT	BRANCH IF BIT 7 OF STATUS REG. IS HIGH

The above instruction will load the content of the controller's status register into the accumulator. Bit 7 of the status register is the write protect lag. A "one" in bit 7 of the status register indicates that a write protected diskette is inserted into the drive. A BMI instruction will check the write protect flag. The program will branch to the WRPROT address label if the write protect flag is set.

READ A DATA BYTE

LDA LDA BPL	Q7L, X Q6L, X LOOP	MAKE SURE IN READ MODE READ THE BYTE STAY IN THE LOOP IF THE M.S. BIT IS LOW
•		
	061 Y	DE LD ANOMNED DUMB
BPL .	LP	READ ANOTHER BYTE SAY IN THE LOOP IF THE M.S. BIT IS LOW
	LDA BPL LDA BPL	LDA Q6L, X BPL LOOP LDA Q6L, X BPL LP

NOTE: THERE SHOULD BE NO PAGE CROSSING FOR THE BPL INSTRUCTIONS.

The LDA Q7L, X instruction makes sure that the controller is in read mode. The LDA Q6L, X instruction loads the contents of the controller's data shift register into the accumulator. Since the Apple GCR code requires that the most significant bit of every encoded data byte is high, the BPL instruction will force the program to stay in a two instruction loop until the M.S. bit of the controller data shift register is high. At the beginning of every byte time, the controller internal logic will clear the data shift register. As data bits are shifted into the data shift register, the M.S. bit of the register is high, if and only if a full byte of data is assembled in the register. The data byte will stay in the register for a little more than 7 us. Therefore, it is important to make sure that the BPL instruction does not cross the page boundary. This is necessary to ensure that the execution time of the two instruction loop (LDA, BPL) is no more than 7 us.

WRITE A DATA BYTE

			LDA	Q6H,	X	GO TO
			LDA	Q7L,		PREWRITE STATE
		•	LDA	DATA		•
			STA	Q7H,	X	PARALLEL LOAD DATA INTO CONTROLLER
			LDA	Q6L,		CONTROLLER SHIFT DATA OUT SERIALLY
		EXECUTION TIME	•	, ,		
	OF	THESE INSTRUCTION	s .			
		MUST BE EXACTLY	•			
		32 CLOCK CYCLES	•			
١			STA	Q6H,	X	PARALLEL LOAD ANOTHER BYTE
			LDA	Q6L,	X	SHIFT OUT DATA
			•			•
			•			
			•			
			LDA	Q7L,	X	OUT OF WRITE MODE
			LDA	Q6L,	X	TO READ MODE
		•				

NOTE: It is important to write a garbage byte (HEX FF) before turning off the write mode, so that the drive electronics has enough time to write the last valid data byte.

The first two instructions force the controller into the prewrite state. These are the same instructions to sense the write protect flag. It is important to execute these two instructions even the programmer does not want to sense the write protect flag. The STA instruction loads the contents of the accumulator into the controller's data shift register. The next instruction [LDA Q6L,'X] causes the data in the register to shift out serially. Q6H and Q7H are the conditions required for parallel loading the data into the controller. Shifting out the data serially to the disk drive requires Q6L and Q7H. The first STA instruction sets Q7 high, because the conditions are Q6H and Q7L before executing this instruction. The conditions are Q6L and Q7H before the second STA instruction, therefore the second STA instruction sets Q6H.

The execution time of the instructions between the end of two consecutive parallel load instructions [STA] has to be exactly 32 clock cycles, otherwise invalid data will be written on the diskette. In order to calculate the execution time, it is important to note that the 6502 processor requires one additional execution cycle for branching or indexing operations crossing the page boundary. The program should switch the controller back to the read mode after all the data has been written.

WRITE SELF SYNC BYTE

			LDA LDA LDA	Q6H, Q7L, #\$FF	X	
		***	STA	Q7H,		PARALLEL LOAD AUTO SYNC BYTE
			ORA	Q6L,	X	START TO SHIFT OUT AUTO SYNC BYTE
l		EXECUTION TIME	•			ORA IS USED SO THAT LDA #\$FF IS NOT
1	OF	THESE INSTRUCTION	s .			NEEDED TO WRITE THE NEXT SYNC CYCLE
		MUST BE EXACTLY 40 CLOCK CYCLES	•			·
1			STA	Q6H,	X	PARALLEL LOAD ANOTHER AUTO SYNC BYTE
•	-		ORA	Q6L,		SHIFT OUT SYNC BYTE
		SELF SYNC BYTE 40 CLOCK CYCLES	•	. ,		•
1			LDA	DATA		
l			STA	Q6H,		LOAD FIRST DATA BYTE
٠			LDA	Q6L,		SHIFT OUT DATA BYTE
		DATA BYTE	•	,		
		32 CLOCK CYCLES	•			
			•			
1			STA	Q6H,	X	
			LDA	Q6L,	X	
			•			
			•			
			•			
			LDA	Q7L,		OUT OF WRITE MODE
			LDA	Q6L,	X	TO READ MODE
ı						

NOTE: Write a garbage byte (HEX FF) before turning off write mode.

The Apple GCR code uses a unique synchronization technique to determine the byte boundary. A self sync byte consists of eight bits of "l" and two bits of "0." The procedure to write a sync byte is the same as to write a data byte, except that the execution time of the instructions between the end of two consecutive parallel load sync byte instructions has to be exactly 40 us. During the write sync byte time, the processor loads eight bits of "l" into the controller. After shifting out 8 bits of "1", the controller hardware will shift out "0" until the next parallel load instruction. Since there are 40 us between two consecutive parallel load instructions and a 4 us bit time, 8 bits of "1" from the processor and 2 bits of "0" appended by the controller hardware are shifted out to the drive. It is necessary to write at least five self sync bytes at the beginning of both the address and data field.

READ SELF SYNC BYTE

Due to the Apple GCR code's unique synchronization technique, the controller hardware will determine the byte boundary automatically. The following is a brief description of the Apple synchronization technique.

FIGURE 1: SYNCHRONIZATION PROCESS

5TH

1ST SYNC BYTE SYNC BYTE

In the above diagram, each row of brackets represent what the controller will send out to the Apple II should the controller start reading at any given bit in the first self sync byte. The controller groups the self sync read data stream into 8-bit byte with a "l" in the most significant bit of each byte. Any "0" bit between bytes are dropped out. From the above diagram, it is shown that the controller is able to group the data at the correct byte boundary within five byte time after the beginning of the read. This is always true for any bit position to start the reading. Therefore, a minimum of five self sync bytes are required for the controller to sync on the read data. After the fifth self sync byte, the controller has established the byte boundary and is able to read the data following the sync bytes correctly. The "D5 AA 96" and "D5 AA AD" address mark sequences follows the self sync bytes in the address and data field respectively. It is not necessary to read and verify the sync byte. In order to read/write a sector, the program should look for the "D5 AA 96" sequence which are the address mark bytes for the address field.

The D5 and AA patterns are reserved for address mark. These patterns are not used to encode data. Therefore, byte syncronization for the address field is achieved by searching for the "D5 AA 96" sequence. Byte synchronization for the data field is done by looking for the "D5 AA AD" sequence.

SEEK TO ANOTHER TRACK

The stepper motor in the Disk II is a four phase stepper motor. Eight I/O control softswitches are used to toggle the four phase on and off as shown in table 2. Two adjacent phases have to be activated in sequence in order to move the R/W head to the adjacent track. Activating the phases in ascending order (0, 1, 2, 3, 0, 1, ...) moves the head towards (inward) the center of the diskette. The head moves away (outward) from the center of the diskette when the phases are activated in descending order (3, 2, 1, 0, 3, 2, ...). All even numbered tracks are positioned under phase 0 and all odd numbered tracks are under phase 2. In order to step in a track, the phase 1 and then phase 2 have to be activated in sequence from an even numbered track, while the phase 3 and then phase 0 is activated in sequence from an odd numbered track. The phase 3 and then phase 2 sequence is used to step out a track from an even numbered track. For stepping out a track from an odd numbered track, the phase 1 and then phase 0 sequence is used. The spindle motor should be on for 150 ms before starting the seek operation. The following is an example to step in a track from an even numbered track.

LDA \$C083, X TURN ON PHASE 1

11.5 msec delay loop

LDA \$C085, X TURN ON PHASE 2


0.1 msec delay loop

LDA \$C084, X TURN OFF PHASE 1

36.6 msec delay loop

LDA \$C086, X TURN OFF PHASE 2

The above programming example is used to illustrate the timing required to step in a track from an even numbered track. The user may use a indexed look up table for the parameter required for different delay loops. No matter how many tracks to step, the user has to allow the last phase to be on for 36.6 msec, because this timing includes the head settling time requirement (25 ms) of the drive. For long seek (step a number of tracks), two adjacent phases can overlap the phase on time in order to increase the torque of the stepper motor and to reduce the seek time. Since the timing between the phase ON/OFF time is critical, it is recommended that the user calls upon the SEEK routine in the Apple DOS for seeking. Figure 2 shows the waveforms of the phases to seek from track 0 to track 9.



NOTE: ALL THE NUMBERS SHOWN IN THE DIAGRAM ARE IN MILLISECONDS.

DATE:

May 29, 1984

TO: Distribution

FROM:

L

1.2

SUBJECT:

Software Routine to Determine A

State Machine or IWM Controller

The attached software routine will determine whether a state machine disk II controller or an IWM controller is installed in the system. This routine may be useful for programmer writing copy protection codes for Apple II.

Upon exiting from the routine, the IWM controller will restore back to synchronous mode (line no. 46 and 47), since all our current applications of IWM for Apple II is in that mode. Future application of the IWM may require the asynchronous mode, the user should set up the mode register (line 46) accordingly.

NL: npk Attachment

DISTRIBUTION

M. Ac

P B

C C C

J. S

R W

```
ĴΕ
         FILE #81 =>IWM
00001
                      1 #
8888:
                      2 *
19991
                      3 * THIS ROUTINE WILL DETERMINE WHETHER A STATE MACHINE DISK II CONTROLL!
8888:
                      4 * OR AN IWM CONTROLLER IS INSTALLED IN THE SYSTEM.
8888:
                      5 * UPON EXIT FROM THE ROUTINE, Y=1 MEANS IWM CONTROLLER AND Y=0 MEANS
                      6 * STATE MACHINE DISK II CONTROLLER.
9999:
                      7 *
9999:
8888:
                      8 * ASSUME A MOTOR OFF INSTRUCTION [LDA $C888,X] HAS BEEN EXECUTED FOR
                      9 * TWO SECONDS BEFORE THE USER CALLS ON THIS ROUTINE.
                                                                                  OTHERWISE, A
8688:
                     18 * TWO SECOND DELAY LOOP MUST BE ADDED AFTER THE FIRST MOTOR OFF
9999:
                     11 * INSTRUCTION [LDA $C088,X] AT THE BEGINNING OF THIS ROUTINE.
0000:
                     12 *
3000:
                     13 * THE ENABLED DISK DRIVE WILL CONTINUE TO BE ON FOR 1 SEC
8888:
                     14 * AFTER EXIT FROM THIS ROUTINE.
8686:
                     15 *
8888:
9999:
                     16 *
---- NEXT OBJECT FILE NAME IS IWM. 0
                                   ORG
                                         $1999
              1000
                     17
1000:
1000:AE 41 10
                                   LDX
                                         SLOTX16
                                                        1X REG=SLOT NO. X 16
                     18
1003:BD 88 C0
                     19
                                   LDA
                                         $C988,X
                                                        :MOTOR OFF
                     28
                                   LDY
                                                        :CLEAR REG. Y
1886:A8 88
                                         #88
                     21
                                   LDA
                                         $C88D,X
                                                        :06H
1998:BD BD C9
                                                        ;Q7H, ADDRESS MODE REG.
                     22
                                   LDA
                                         $C88F,X
1888:BD 8F C8
                     23
                                   LDA
                                         #$84
    :A9 84
     .9D 8F C8
                                         $C08F,X
                                                        :DISABLE TIMER BIT IN MODE REG.
                     24
                                   STA
                                                        ;Q7L, OUT OF WRITE MODE
                     25
                                   LDA
                                         $C08E.X
1813:BD 8E C0
                                   LDA
                                         $C889,X
                                                        :MOTOR ON
                     26
1016:BD 89 C0
                                                         :18 MSEC DELAY LOOP
1919:48
                     27 LOOP
                                   PHA
                                                         :WAIT FOR THE MOTOR ON
                     28
                                   PLA
181A:68
                                                         ISIGNAL TO BE ACTIVE IN
                                   PHA
                     29
181B:48
                                                         THE CONTROLLER CARD
                     38
                                   PLA
181C:68
                     31
                                   PHA
191D:48
                     32
                                   PLA
181E:68
                     33
                                   PHA
181F:48
                     34
                                   PLA
1828:68
                     35
                                   PHA
1821:48
                                   PLA
1822:68
                     36
                     37
                                   INY
1923:C8
                                                        END OF DELAY LOOP
                     38
                                   BNE
                                          LOOP
1024:D0 F3
              1819
                                                        ;Q7L, READ STATUS REG.
                     39
                                   LDA
                                         $C88E,X
1926:BD 8E C9
                                                        MOTOR OFF
                      48
                                   STA
                                          $C088,X
1829:9D 88 C8
                                                        MASK 5 L.S. BITS
                                   AND
                                          #$1F
                      41
192C:29 1F
                                                        :CHECK TIMER BIT
                                          #$84
182E:C9 84
                                   CMP
                      42
                                                        DISK II CONTROLLER IF NOT EQ
              103B
                      43
                                   BNE
                                          DISKII
1830:D8 89
                                                          INCREMENT Y REG.
                                   INY
                      44 IWM
1032:C8
                                                        :Q7H, ADDRESS MODE REG.
                                   LDA
                                          $C08F,X
                      45
1933:BD 8F C8
1836:A9 88
                      46
                                   LDA
                                          #$88
                                                        107H, RESTORE MODE REG.
                                   STA
                                          $C98F,X
1938:9D 8F C9
                      47
                                   LDA
                                          $C08E,X
                                                        ;Q7L,
      BD 8E C8
                      48 DISKII
                                                        :Q6L, RESTORE TO READ MODE
     .8D 8C C8
                      49
                                   LDA
                                          $C08C.X
                                   EQU
1841:
               1841
                      58 FIN
                                          ¥
                                   DFB
                                          $68
                      51 SLOTX16
1941:60
```

185. DISKII 71841 FIN 71832 IWM 1819 LOOP 1841 SLOTX16

** SUCCESSFUL ASSEMBLY := NO ERRORS
* 'SSEMBLER CREATED ON 15-JAN-84 21:28
TAL LINES ASSEMBLED 51

REE SPACE PAGE COUNT 89